
F07 – Linear Equations (LAPACK)

F07UUF (CTPCON/ZTPCON) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F07UUF (CTPCON/ZTPCON) estimates the condition number of a complex triangular matrix, using
packed storage.

2 Specification

SUBROUTINE F07UUF(NORM, UPLO, DIAG, N, AP, RCOND, WORK, RWORK, INFO)
ENTRY ctpcon(NORM, UPLO, DIAG, N, AP, RCOND, WORK, RWORK, INFO)
INTEGER N, INFO
real RCOND, RWORK(∗)
complex AP(∗), WORK(∗)
CHARACTER∗1 NORM, UPLO, DIAG

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number of a complex triangular matrix A, in either the 1-norm or
the infinity-norm, using packed storage:

κ1(A) = ‖A‖1‖A−1‖1 or κ∞(A) = ‖A‖∞‖A−1‖∞.

Note that κ∞(A) = κ1(A
T ).

Because the condition number is infinite if A is singular, the routine actually returns an estimate of the
reciprocal of the condition number.

The routine computes ‖A‖1 or ‖A‖∞ exactly, and uses Higham’s implementation of Hager’s method [1]
to estimate ‖A−1‖1 or ‖A−1‖∞.

4 References

[1] Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with
applications to condition estimation ACM Trans. Math. Software 14 381–396

5 Parameters

1: NORM — CHARACTER*1 Input

On entry: indicates whether κ1(A) or κ∞(A) is estimated as follows:

if NORM = ’1’ or ’O’, then κ1(A) is estimated;
if NORM = ’I’, then κ∞(A) is estimated.

Constraint: NORM = ’1’, ’O’ or ’I’.

2: UPLO — CHARACTER*1 Input

On entry: indicates whether A is upper or lower triangular as follows:

if UPLO = ’U’, then A is upper triangular;
if UPLO = ’L’, then A is lower triangular.

Constraint: UPLO = ’U’ or ’L’.

3: DIAG — CHARACTER*1 Input

On entry: indicates whether A is a non-unit or unit triangular matrix as follows:

if DIAG = ’N’, then A is a non-unit triangular matrix;
if DIAG = ’U’, then A is a unit triangular matrix; the diagonal elements are not referenced
and are assumed to be 1.

Constraint: DIAG = ’N’ or ’U’.

4: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

5: AP(∗) — complex array Input

Note: the dimension of the array AP must be at least max(1,N∗(N + 1)/2).

On entry: the n by n triangular matrix A, packed by columns. More precisely, if UPLO = ’U’, the
upper triangle of A must be stored with element aij in AP(i + j(j − 1)/2) for i ≤ j;if UPLO = ’L’,
the lower triangle of A must be stored with element aij in AP(i + (2n − j)(j − 1)/2) for i ≥ j. If
DIAG = ’U’, the diagonal elements of the matrix are not referenced and are assumed to be 1;the
same storage scheme is used whether DIAG = ’N’ or ’U’.

6: RCOND — real Output

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact
singularity is detected or the estimate underflows. If RCOND is less than machine precision,
then A is singular to working precision.

7: WORK(∗) — complex array Workspace

Note: the dimension of the array WORK must be at least max(1,2∗N).

8: RWORK(∗) — real array Workspace

Note: the dimension of the array RWORK must be at least max(1,N).

9: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less
than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form Ax = b or
AHx = b; the number is usually 5 and never more than 11. Each solution involves approximately 4n2

real floating-point operations but takes considerably longer than a call to F07USF (CTPTRS/ZTPTRS)
with one right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The real analogue of this routine is F07UGF (STPCON/DTPCON).

9 Example

To estimate the condition number in the 1-norm of the matrix A, where

A =




4.78 + 4.56i 0.00 + 0.00i 0.00 + 0.00i 0.00 + 0.00i
2.00− 0.30i −4.11 + 1.25i 0.00 + 0.00i 0.00 + 0.00i
2.89− 1.34i 2.36− 4.25i 4.15 + 0.80i 0.00 + 0.00i

−1.89 + 1.15i 0.04− 3.69i −0.02 + 0.46i 0.33− 0.26i


 ,

using packed storage. The true condition number in the 1-norm is 70.27.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F07UUF Example Program Text
* Mark 15 Release. NAG Copyright 1991.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX
PARAMETER (NMAX=8)
CHARACTER NORM, DIAG
PARAMETER (NORM=’1’,DIAG=’N’)

* .. Local Scalars ..
real RCOND
INTEGER I, INFO, J, N
CHARACTER UPLO

* .. Local Arrays ..
complex AP(NMAX*(NMAX+1)/2), WORK(2*NMAX)
real RWORK(NMAX)

* .. External Functions ..
real X02AJF
EXTERNAL X02AJF

* .. External Subroutines ..
EXTERNAL ctpcon

* .. Executable Statements ..
WRITE (NOUT,*) ’F07UUF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read A from data file
*

READ (NIN,*) UPLO
IF (UPLO.EQ.’U’) THEN

READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
ELSE IF (UPLO.EQ.’L’) THEN

READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
END IF

*
* Estimate condition number
*

CALL ctpcon(NORM,UPLO,DIAG,N,AP,RCOND,WORK,RWORK,INFO)
*

WRITE (NOUT,*)
IF (RCOND.GE.X02AJF()) THEN

WRITE (NOUT,99999) ’Estimate of condition number =’,
+ 1.0e0/RCOND

ELSE
WRITE (NOUT,*) ’A is singular to working precision’

END IF
END IF
STOP

*
99999 FORMAT (1X,A,1P,e10.2)

END

9.2 Program Data

F07UUF Example Program Data
4 :Value of N
’L’ :Value of UPLO

( 4.78, 4.56)
( 2.00,-0.30) (-4.11, 1.25)
( 2.89,-1.34) ( 2.36,-4.25) ( 4.15, 0.80)
(-1.89, 1.15) ( 0.04,-3.69) (-0.02, 0.46) ( 0.33,-0.26) :End of matrix A

9.3 Program Results

F07UUF Example Program Results

Estimate of condition number = 3.74E+01

